Maple-assisted proof of empirical formula for A224739

A 2×2 $0 \ldots 1$ matrix can have permanent 0, 1 or 2. There are 38 possible 2×3 matrices:

```
> M:= select(m -> m[1,1]*m[2,2]+m[2,1]*m[1,2] = m[1,2]*m[2,3]+m[2,2]*m[1,3], [seq(seq(seq(seq(seq(seq([[m[1,1],m[1,2],m[1,3]],[m[2,1],m[2,2],m[2,3]]], m[1,1]=0..1),m[1,2]=0..1),m[1,3]=0..1),m[2,1]=0..1),m[2,2]=0..1),m[2,3]=0..1)]);
  nops(M23);
```

M := [[[0, 0, 0], [0, 0, 0]], [[1, 0, 0], [0, 0, 0]], [[0, 1, 0], [0, 0, 0]], [[1, 1, 0], [0, 0, 0]], [[0, 0, 1], [0, 0, 0]], [[1, 0, 1], [0, 0, 0]], [[0, 1, 1], [0, 0, 0]], [[1, 1, 1], [0, 0, 0]], [[0, 0, 0], [1, 0, 0]], [[1, 0, 0], [1, 0, 0]], [[0, 0, 1], [1, 0, 0]], [[1, 0, 1], [1, 0, 0]], [[0, 0, 0], [0, 1, 0]], [[0, 1, 0], [0, 1, 0]], [[1, 0, 1], [0, 1, 0]], [[1, 1, 1], [0, 1, 0]], [[0, 0, 0], [1, 1, 0]], [[1, 0, 1], [1, 1, 0]], [[0, 1, 1], [1, 1, 0]], [[0, 0, 0], [0, 0, 1]], [[1, 0, 0], [0, 0, 1]], [[0, 0, 1], [0, 0, 1]], [[1, 0, 1], [0, 0, 1]], [[0, 0, 0], [1, 0, 1]], [[1, 0, 0], [1, 0, 1]], [[0, 1, 0], [1, 0, 1]], [[1, 1, 0], [1, 0, 1]], [[0, 0, 1], [1, 0, 1]], [[1, 0, 1], [1, 0, 1]], [[0, 1, 1], [1, 0, 1]], [[1, 1, 1], [1, 0, 1]], [[0, 0, 0], [0, 1, 1]], [[1, 1, 0], [0, 1, 1]], [[1, 0, 1], [0, 1, 1]], [[0, 0, 0], [1, 1, 1]], [[0, 1, 0], [1, 1, 1]], [[1, 0, 1], [1, 1, 1]], [[1, 1, 1], [1, 1, 1]]]

38 **(1)**

Let T be the 38×38 matrix such that $T_{ij} = 1$ if the top two rows of a 3×3 matrix could be M_i and the bottom two rows M_j, the middle row being common to both, and all 2×2 permanents equal.

```
> T:= Matrix(38,38, proc(i,j) if M[i][2] = M[j][1] and M[i][1,1]*M[i][2,2]+M[i][1,2]*M[i][2,1] = M[j][1,1]*M[j][2,2]+M[j][1,2]*M[j][2,1] then 1 else 0 fi end proc):
```

Then $a_n = Trace(T^{n-1}) = e^T T^{n-1} e$ where e is the column vector of all 1's. To check, here are the first few members of the sequence.

```
> e:= Vector(38,1): Tu[0]:= e:
  for i from 1 to 30 do Tu[i]:= T . Tu[i-1] od:
> seq(e^%T . Tu[i], i=0..30);
```

38, 152, 565, 2326, 9554, 40297, 170754, 728996, 3120401, 13387658, 57499978, 247151833, 1062764258, 4571134864, 19664166357, 84599301422, 363983008394, 1566063674345, 6738231845242, 28992616966540, 124747487937041, 536757282833410, 2309538038609730, 9937403210056409, 42758363151913178, 183979506812077592, 791622190153982005, 3406171778623300486, 14655990653131948034, 63061433653454421097, 271339185353421712434 **(2)**

Here is the minimal polynomial of T.

```
> P:=LinearAlgebra:-MinimalPolynomial(T,z);
```

$$P := z^{10} - 6z^9 + 41z^7 - 25z^6 - 86z^5 + 60z^4 + 51z^3 - 36z^2 \quad \textbf{(3)}$$

Thus a_n must satisfy the linear recurrence of order 10 corresponding to P. However, it seems it satisfies the empirical recurrence of order 7, corresponding to the following polynomial $Q(z)$.

```
> Q:= z^7 - 7*z^6 + 7*z^5 + 34*z^4 - 59*z^3 - 27*z^2 + 87*z - 36;
```

$$Q := z^7 - 7z^6 + 7z^5 + 34z^4 - 59z^3 - 27z^2 + 87z - 36 \quad \textbf{(4)}$$

We check that $Q(z)$ divides $P(z)$, and let the quotient be $R(z)$.

```
> R:= normal(P/Q);
```

$$R := (z+1)\, z^2 \tag{5}$$

If

$b_n = a_n - 7\ a_{n-1} + 7\, a_{n-2} + 34\, a_{n-3} - 59\, a_{n-4} - 27\, a_{n-5} + 87\, a_{n-6} - 36\, a_{n-7} = e^T Q(T)\ T^{n-8}\, e$
for $n \geq 8$, then b_n satisfies the recurrence of order 3 corresponding to $R(z)$, i.e. $b_{n+3} + b_{n+2} = 0$.
To show that a_n satisfies the empirical recurrence for all $n \geq 8$, it suffices to check $b_8 = b_9 = b_{10} = 0$.

```
> seq(e^%T . (Tu[n-1] - 7*Tu[n-2]+7*Tu[n-3]+34*Tu[n-4]-59*Tu[n-5]
  -27*Tu[n-6]+87*Tu[n-7]-36*Tu[n-8]), n=8..10);
```

$$0, 0, 0 \tag{6}$$